unittest
{
	assert( [ "2 50", "6 10" ].parse.expand.solve == 2 );
	assert( [ "3 100", "14 22 40" ].parse.expand.solve == 0 );
	assert( [ "5 1000000000", "6 6 2 6 2" ].parse.expand.solve == 166666667 );
	
	assert( [ "2 1000000000", "2 4" ].parse.expand.solve == 0 );		// 追加ケース
}



import std.conv;
import std.numeric;
import std.range;
import std.stdio;
import std.typecons;

void main()
{
	stdin.byLineCopy.parse.expand.solve.writeln;
}

auto parse( Range )( Range input )
if( isInputRange!Range && is( ElementType!Range == string ) )
{
	auto nm = input.front.split.to!( long[] );
	input.popFront;
	auto as = input.front.split.to!( long[] );
	return tuple( nm[ 1 ], as );
}

auto solve( long m, long[] as )
{
	// aの半分の最小公倍数（＝最小のX候補）を計算
	auto x = 1L;
	foreach( a; as )
	{
		if( a % 2 ) return 0;		// aが奇数だったらpを何にしてもXが整数にならないので0確定
		x = lcm( x, a / 2 );
		if( m < x ) return 0;		// mより大きかったら0確定
	}
	
	// aの半分の最小公倍数が半公倍数じゃなかったら0確定
	foreach( a; as )
	{
		if( ( x - a / 2 ) % a ) return 0;
	}
	
	// 0 < i && x * i <= m && i % 2 == 1 のときに半公倍数になる
	return ( m / x + 1 ) / 2;
}

// 最小公倍数
auto lcm( long x, long y )
{
	return x * y / x.gcd( y );
}
